Sincronizzazione in RG

Metabolizzazione: fase 1

Introduzione

Sono costretto ad usare l'indice 4 invece di 0 per l'indice della coordinata temporale, a causa delle convenzioni matriciali di *Mathematica*. Inoltre, c=1.

Elemento ds^2

```
Clear[g, dx, dy, dz, dt, dT];
g[i_, j_] := g[j, i] /; j > i  (* regola di simmetria per gij *)
Format[g[i_, j_]] := g_{i,j}
```

Elemento di lunghezza invariante al quadrato:

```
ds2[ g_, dr_] := Sum[ g[i, j] * dr[[i]] * dr[[j]], {i, 1, 4}, {j, 1, 4}]
```

```
ds2[g, {dX, dY, dZ, dT}] // Expand
```

$$dX^2\, g_{1,1} + 2\, dX\, dY\, g_{2,1} + dY^2\, g_{2,2} + 2\, dX\, dZ\, g_{3,1} + 2\, dY\, dZ\, g_{3,2} + dZ^2\, g_{3,3} + 2\, dT\, dX\, g_{4,1} + 2\, dT\, dY\, g_{4,2} + 2\, dT\, dZ\, g_{4,3} + dT^2\, g_{4,4}$$

Propagazione lampi di sincronizzazione

Abbiamo due linee di universo: una per x=0 e una infinitamente vicina x=dx. Consideriamo l'evento nell'origine A(0,0), e l'evento B(dx,0). Sulla linea x=dx siamo interessati anche ad altri due eventi di tipo B(dx,dt): uno con dt>0, l'altro con dt<0.

Da $B_1 = (\mathrm{dx}, -\mathrm{dt}_1)$ parte un lampo luminoso alla volta di A, e vi arriva a t=0; in A(0,0) rimbalza e ritorna in $B_2 = (\mathrm{dx}, +\mathrm{dt}_2)$. I tempi dt_1 e dt_2 corrispondono alle soluzioni dell'eq. di secondo grado ds2=0, dato che, per definizione, la luce di muove con ds=0.

```
Clear[dt]
sol = Solve[ ds2[g, {dx, dy, dz, dt}] == 0, dt];
{sol1 = sol[[1]] // First,  sol2 = sol[[2]] // First};
dt_1 = - dt /. sol1; dt_2 = dt /. sol2;
```

Ecco le due soluzioni (la prima ha il segno − davanti alla radice).

```
{-dt1, dt2}
```

$$\left\{\frac{1}{2\,g_{4,4}}\left(-2\,dx\,g_{4,1}-2\,dy\,g_{4,2}-2\,dz\,g_{4,3}-\sqrt{\left((2\,dx\,g_{4,1}+2\,dy\,g_{4,2}+2\,dz\,g_{4,3})^2-4\,(dx^2\,g_{1,1}+2\,dx\,dy\,g_{2,1}+dy^2\,g_{2,2}+2\,dx\,dz\,g_{3,1}+2\,dy\,dz\,g_{3,2}+dz^2\,g_{3,3})\,g_{4,4}\right)}\right),\right.$$
$$\left.\frac{1}{2\,g_{4,4}}\left(-2\,dx\,g_{4,1}-2\,dy\,g_{4,2}-2\,dz\,g_{4,3}+\sqrt{\left((2\,dx\,g_{4,1}+2\,dy\,g_{4,2}+2\,dz\,g_{4,3})^2-4\,(dx^2\,g_{1,1}+2\,dx\,dy\,g_{2,1}+dy^2\,g_{2,2}+2\,dx\,dz\,g_{3,1}+2\,dy\,dz\,g_{3,2}+dz^2\,g_{3,3})\,g_{4,4}\right)}\right)\right\}$$

Sono uguali?

Affinche' sia $dt_1 = dt_2$ dev'essere $-dt_1 + dt_2 = 0$ (soluzioni uguali e opposte), e questo dovrebbe aversi quando il termine di primo grado in dx (che rappresenta la somma delle soluzioni) e' nullo, cosa che non e' in generale vera. Che tempo dt_0 corrisponde al punto medio tra B_1 e B_2?

```
dt0 = (-dt1 + dt2) / 2 // Expand // Simplify
```

$$-\frac{dx\,g_{4,1}+dy\,g_{4,2}+dz\,g_{4,3}}{g_{4,4}}$$

L'evento $B_0 = (dx, dt_0)$ e' l'evento sulla linea di universo di B *che va considerato come contemporaneo* con l'evento A(0,0) (Landau II, p 311). Quando l'orologio di A segna t=0, quello di B deve segnare $t = dt_0$. [*ma non sono sicuro di quello che dico*]

Tempo di andata e ritorno del segnale

Il tempo per "andare e tornare" del raggio luminoso (in coordinate universali) e' la somma dei due (in valore assoluto) (Landau II, p308):

```
dt = dt1 + dt2  // Expand
```

$$\frac{1}{g_{4,4}}\left(\sqrt{\left((2\,dx\,g_{4,1}+2\,dy\,g_{4,2}+2\,dz\,g_{4,3})^2-4\,(dx^2\,g_{1,1}+2\,dx\,dy\,g_{2,1}+dy^2\,g_{2,2}+2\,dx\,dz\,g_{3,1}+2\,dy\,dz\,g_{3,2}+dz^2\,g_{3,3})\,g_{4,4}\right)}\right)$$

Per esprimere questo dt in "coordinate locali" in x=0, occorre introdurre il tempo proprio: $d\tau^2 = g_{00}\,dt^2$. Per trovare $d\tau$, basta quindi moltiplicare per la radice di g00:

```
dτ = Sqrt[g[4, 4]] * dt
```

$$\frac{1}{\sqrt{g_{4,4}}}\left(\sqrt{\left((2\,dx\,g_{4,1}+2\,dy\,g_{4,2}+2\,dz\,g_{4,3})^2-4\,(dx^2\,g_{1,1}+2\,dx\,dy\,g_{2,1}+dy^2\,g_{2,2}+2\,dx\,dz\,g_{3,1}+2\,dy\,dz\,g_{3,2}+dz^2\,g_{3,3})\,g_{4,4}\right)}\right)$$

Distanza spaziale in RG

Per definire la distanza spaziale dL (cioe' la metrica nello spazio R^3 indotta dalla gravita'), si sfrutta il fatto che la velocita' della luce e' sempre c, quindi $dL = c * \frac{d\tau}{2}$:

```
dL = dτ / 2 // Expand
```

$$\frac{1}{2\sqrt{g_{4,4}}}\left(\sqrt{\left((2\,dx\,g_{4,1}+2\,dy\,g_{4,2}+2\,dz\,g_{4,3})^2-4\,(dx^2\,g_{1,1}+2\,dx\,dy\,g_{2,1}+dy^2\,g_{2,2}+2\,dx\,dz\,g_{3,1}+2\,dy\,dz\,g_{3,2}+dz^2\,g_{3,3})\,g_{4,4}\right)}\right)$$

Tensore metrico tridimensionale γ

Per definizione γ e' la matriche che permette di scrivere $dL^2 = dx*\gamma*dx$

```
Clear[γ]
Format[γ[i_, j_]] := γi,j
dL2 = dL^2 // Expand
```

$$-dx^2\,g_{1,1}-2\,dx\,dy\,g_{2,1}-dy^2\,g_{2,2}-2\,dx\,dz\,g_{3,1}-2\,dy\,dz\,g_{3,2}-dz^2\,g_{3,3}+\frac{dx^2\,g_{4,1}^2}{g_{4,4}}+\frac{2\,dx\,dy\,g_{4,1}\,g_{4,2}}{g_{4,4}}+\frac{dy^2\,g_{4,2}^2}{g_{4,4}}+\frac{2\,dx\,dz\,g_{4,1}\,g_{4,3}}{g_{4,4}}+\frac{2\,dy\,dz\,g_{4,2}\,g_{4,3}}{g_{4,4}}+\frac{dz^2\,g_{4,3}^2}{g_{4,4}}$$

Come esempio, consideriamo un termine misto 12 (quello che moltiplica dx*dy)

```
γ[g, 1, 2] = 1/2 * Coefficient[ dL2, dx * dy] // Simplify
```

$$-g_{2,1}+\frac{g_{4,1}\,g_{4,2}}{g_{4,4}}$$

(si divide per due a causa della simmetria) E' facile generalizzare a tutti i termini, scrivendo:

```
γ[g_, i_, j_] = -g[i, j] + g[4, i] * g[4, j] / g[4, 4];
```

(Landau II, p 309)

Esempio numerico: campo rotante

```
Clear[dt]
```

Considero una circonferenza di raggio r: fisso tutte le coordinate (dr=0, dz=0), tranne l'angolo $d\phi$ (che viene assunta come prima coordinata dx) e il tempo dt (che e' la quarta coordinata). Inoltre, queste grandezze vanno considerate relative al corpo che ruota con velocita' angolare ω. La distanza tra due eventi sul bordo del cerchio, che distino $d\phi$ in coordinate locali (e quindi $d\phi+\omega*dt$ in coordinate assolute, vale $ds^2 = dt^2 - r^2\,(d\phi+\omega dt)^2$

```
q=dt^2 - r^2*(dϕ + ω*dt)^2//Expand
```

$$dt^2-d\phi^2\,r^2-2\,dt\,d\phi\,r^2\,\omega-dt^2\,r^2\,\omega^2$$

E' facile estrarre il tensore metrico:

```
Clear[G];
G[i_, j_] = 0;
G[4, 4] = Coefficient[q, dt^2];
G[1, 1] = Coefficient[q, dϕ^2];
G[1, 4] = Coefficient[q, dt*dϕ] / 2;
G[4, 1] = G[1, 4];
```

Ecco come appare il tensore metrico G:

```
Table[G[i, j], {i, 1, 4}, {j, 1, 4}] // MatrixForm
```

$$\begin{pmatrix} -r^2 & 0 & 0 & -r^2\omega \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -r^2\omega & 0 & 0 & 1-r^2\omega^2 \end{pmatrix}$$

... verifica sull'elemento di linea:

```
ds2[G, {dϕ, 0, 0, dt}]
```

$$-d\phi^2\, r^2 - 2\, dt\, d\phi\, r^2\, \omega + dt^2\, (1 - r^2\omega^2)$$

(Landau II, p 333)

Ecco come appare il tensore tridimensionale γ:

```
Table[γ[G, i, j], {i, 1, 3}, {j, 1, 3}] // Simplify // MatrixForm
```

$$\begin{pmatrix} \frac{r^2}{1-r^2\omega^2} & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

Il termine che sopravvive (11) e' interessante. La sua radice quadrata fornisce:

```
R = Sqrt[γ[G, 1, 1]] // Simplify
```

$$\sqrt{\frac{r^2}{1 - r^2\omega^2}}$$

e che non e' altro che la derivata dL/dϕ. Integrando tra [0,2π] si la la lunghezza della circonferenza in coordinate "rotanti": 2π*R. Sviluppando in serie per piccole ω si ha:

```
2 π * R + O[ω]^5 // PowerExpand
```

$$2\pi r + \pi r^3\omega^2 + \frac{3}{4}\pi r^5\omega^4 + O[\omega]^5$$

Ne consegue che la circonferenza *appare piu' lunga* di 2πr.

(Landau II,p 334)